# Maintainer: Lukas Fleischer <archlinux at cryptocrack dot de>
# Contributor: Allan McRae <allan@archlinux.org>
# Contributor: bekks <eduard.warkentin@gmail.com>
# Contributor: Daniel J Griffiths <ghost1227@archlinux.us>

pkgname=upx
pkgver=3.09
pkgrel=1
pkgdesc='Ultimate executable compressor.'
arch=('i686' 'x86_64')
url='http://upx.sourceforge.net/'
license=('GPL')
depends=('zlib')
makedepends=('ucl')
source=("http://upx.sourceforge.net/download/${pkgname}-${pkgver}-src.tar.bz2"
        'http://downloads.sourceforge.net/sevenzip/lzma465.tar.bz2')
md5sums=('1253da46eac54a217eb73e2d44818e53'
         '29d5ffd03a5a3e51aef6a74e9eafb759')

build() {
  cd "${pkgname}-${pkgver}-src"
  make UPX_LZMA_VERSION=0x465 UPX_LZMADIR="${srcdir}" all
}

package() {
  cd "${pkgname}-${pkgver}-src"

  install -Dm0755 src/upx.out "${pkgdir}/usr/bin/upx"
  install -Dm0644 doc/upx.1 "${pkgdir}/usr/share/man/man1/upx.1"
}
